Urban Computation

Application of basic operations research methods to solve Urban Management and Planning

101-0507-00 G

(ECTS-3KP)

Dr. Nam LETHANH

Zürich FS 2015

ETH Zürich

Institut für Bau- und Infrastrukturmanagement
Professur für Infrastrukturmanagement

Disclaimer:

This document has been compiled to provide students taking the above mentioned course in understanding the examples given in class and assignments. All questions presented have been prepared by the authors based to varying degrees on different sources. The sources are given for each example. This document is being issued so that students taking the course may better prepare for the exam. As it has not yet undergone the complete rigorous control processes of IBI this document is to be considered a work in progress and is hence labeled as a DRAFT. Some of the examples have been modified from those found in

Ragsdale, C.T., (2008), Spreadsheet Modelling and decision analysis: A practical introduction to management science, 5th edition, 820 pages.

Table of contents

1 LO - Salt and sand ordering problem

2 LO – Road resurfacing problem

An infrastructure management company is awarded a contract to do resurfacing of 20 km of road in a high-density residential area. According to the environmental regulation, all activities must be in compliance with environmental regulation. This regulation was enforced in order to protect the stakeholders against environmental issues such as pollution, energy consumption, etc. These regulations mean that the total CO_2 emission from the intervention cannot exceed 13 tons.

In addition to the environmental regulations you are know that the intervention cannot cost more than 2.86 million mu, or otherwise you will lose money, and that you have to have the project completed within 25 days.

You are considering using two different techniques to resurface the road 1) Hot mixed asphalt concrete (HMAC); and 2) Warm mixed asphalt concrete (WMAC). The first one is faster but produces more CO_2 than the second one. It is known that producing and paving 1 m^3 of HMAC discharges approximately 2000 kg of CO_2, and that producing and paving 1 m^3 of WMAC discharges 1600 kg of CO_2.

The maximum amount of HMAC that can be produced and paved is 550 tons/days and the maximum amount of WMAC is 450 tons/day. Pavement with HMAC is 5 cm thick. Pavement with WMAC is 5 cm thick. The road is 6.5 m wide.

The cost to produce and pave HMAC and WMAC are 20 and 25 mu/m^2. The profit from you expect from each is 3 and 4 mu/m^3, respectively.

2.1 Question

How much of the 20 km road section should you pave with HMAC and WMAC to maximize your profit? How many days will you produce and pave with HMAC and WMAC?

2.2 Answer

The answer to this question can be found by realizing that that the amount of road to be paved with each type of asphalt can be modelled as a production problem, since the amount of profit to be made can be linked directly to how much can be produced and paved.

2.2.1 Mathematical formulation

2.2.1.1 Specific

Objective function:

$$\text{Maximize: } Z = \frac{1'000}{2'400 \cdot 0.05}\left[x_1 \cdot 3 \cdot 550 + x_2 \cdot 4 \cdot 450\right] \tag{1}$$

$$\Leftrightarrow Z = 13'750 \cdot x_1 + 15'000 \cdot x_2$$

x_1 and x_2 is the time to produce and pave asphalt type HMAC and WMAC, respectively

Subject to

Functional constraints

(1) Quantity constraint:

$$\left[x_1 \cdot 550 + x_2 \cdot 450\right] = (20 \cdot 1'000) \cdot 0.05 \cdot 6.5 \cdot \frac{2'400}{1'000} = 15'600 \tag{2}$$

(2) Emission constraint:

$$\left[\frac{x_1 \cdot 550 \cdot \left(\frac{2'000}{1'000}\right) + x_2 \cdot 450 \cdot \left(\frac{1'600}{1'000}\right)}{\frac{2'400}{1'000}}\right] \leq 13 \cdot 1'000 \tag{3}$$

$$\Leftrightarrow x_1 \cdot 0.4583 + x_2 \cdot 0.3 \leq 13$$

(3) Budget constraint:

$$\left[\frac{x_1 \cdot 550 \cdot 20 + x_2 \cdot 450 \cdot 25}{\frac{5}{100} \cdot \frac{2'400}{1'000}}\right] \leq 2'860'000 \tag{4}$$

$$\Leftrightarrow 91'667 \cdot x_1 + 93'750 \cdot x_2 \leq 2'860'000$$

(4) Time constraint: $x_1 \leq 20$ (5)

$x_2 \leq 25$ (6)

Non-Negativity constraints

Non-negativity constraints $x_1, x_2 \geq 0$ (7)

2.2.2 Graphical method

Using the graphical method, it can be determined that the optimal numbers of days to produce and pave with HMAC and WMAC asphalt types are approximately 17 days and 14 days, respectively (Figure 1). Only the final graph is shown.



Figure 1: Selection of amount of asphalt type

2.2.3 Solver

2.2.3.1 Model

```
#Model file: IMM-Hot-Warm-asphalt.mod
set asphalt;  # Hot or warm asphalt construction methods
#----pavement information
param thickness; # in cm
param length; # in km
param width; # in m
param density; # kg/m3
param CO2_limit; #allowance level of CO2 (tons)
param CO2{asphalt}; # kg/m3 of asphalt
param rate{asphalt}; # tons/day
param cost{asphalt}; # CHF/m2
param cost_limit;
param time_limit{asphalt}; # days (maximum number of days for intervention)
param profit{asphalt}; # earning per m2 of asphalt
#--derived parameters if needed
param demand := (length*width*thickness*1000)*(density/1000);
#variables
var time{k in asphalt} >=0; # Number of days
#objective function
maximize profits: sum{k in asphalt} time[k]*rate[k]*profit[k]/(thickness*density/1000);
subject to totaldemand:  sum {k in asphalt} time[k]*rate[k] = demand;
subject to emission:  sum {k in asphalt} (time[k]*rate[k])*(CO2[k]/1000)/(density/1000)/1000 <= CO2_limit;
subject to budget: sum{k in asphalt} ((time[k]*rate[k])/(thickness*density/1000))*cost[k] <= cost_limit;
subject to times {k in asphalt}: time[k] <= time_limit[k];
```

2.2.3.2 Data

```
#Data file: IMM-Hot-Warm-asphalt.dat
```

```
data;
set asphalt := HMAC WMAC ; #
param thickness := 0.05; # in cm
param length := 20; # in km
param width := 6.5; # in m
param density := 2400; # kg/m3
param CO2_limit := 13; #allowance level of CO2 (tons)
param cost_limit := 2860000; #CHF
param   CO2:=
        HMAC   2000
        WMAC   1600;# kg/m3 of asphalt
param   rate:=
        HMAC   550
        WMAC   450;# tons/day
param   cost:=
        HMAC   20
        WMAC   25;# CHF/m2
param   time_limit:=
        HMAC   20
        WMAC   25;# days
param   profit:=
        HMAC   3
        WMAC   4;# CHF/m2
```

2.2.4 Results

The optimal strategy is to produce and pave HMAC and WMAC asphalt for 17 days and 14 days, respectively. Following this strategy would result in a profit of 442’000 mu.

2.2.4.1 Results from solver

```
Presolve eliminates 2 constraints.
Adjusted problem:
2 variables, all linear
3 constraints, all linear; 6 nonzeros
   1 equality constraint
   2 inequality constraints
1 linear objective; 2 nonzeros.

Gurobi 5.5.0: outlev=1
threads=4
Optimize a model with 3 rows, 2 columns and 6 nonzeros
Presolve removed 3 rows and 2 columns
Presolve time: 0.00s
Presolve: All rows and columns removed
Iteration    Objective       Primal Inf.    Dual Inf.      Time
       0    4.4200000e+05   0.000000e+00   0.000000e+00      0s

Solved in 0 iterations and 0.00 seconds
Optimal objective  4.420000000e+05
Gurobi 5.5.0: optimal solution; objective 442000
time [*] :=
HMAC  17.0182
WMAC  13.8667
```

3 ILO - Power generation problem

As the owner of a power company you are considering how to increase your generating capacity to meet expected demand in your growing service area. Currently, you have 750 MW of generating capacity but predict you will need the minimum generating capacities in each of the next five years shown in Table 2.

Table 2. Minimum required generating capacity

Year	1	2	3	4	5
Capacity (MW)	780	860	950	1‘060	1‘180

You can increase your generating capacity by purchasing generators of four different types: 10 MW, 25MW, 50MW and/or 100MW. The cost of acquiring and installing a generator of each type in the next five years is given in Table 3:

Table 3. Cost of generators

Generator type	Costs of generator (in 1’000 mu) in year				
	1	2	3	4	5
10 MW	300	250	200	170	145
25MW	460	375	350	280	235
50MW	670	558	465	380	320
100MW	950	790	670	550	460

3.1 Question

How many generators of each type should you buy and install each year to meet demand but to minimize purchasing cost?

3.2 Answer

3.2.1 Mathematical formulation

3.2.1.1 Specific

Objective function:

Minimize purchasing cost:

$$\begin{aligned} Z = {} & 300 \cdot x_{1,1} + 460 \cdot x_{2,1} + 670 \cdot x_{3,1} + 950 \cdot x_{4,1} + \\ & 250 \cdot x_{1,2} + 375 \cdot x_{2,2} + 558 \cdot x_{3,2} + 790 \cdot x_{4,2} + \\ & 200 \cdot x_{1,3} + 350 \cdot x_{2,3} + 465 \cdot x_{3,3} + 670 \cdot x_{4,3} + \\ & 170 \cdot x_{1,4} + 280 \cdot x_{2,4} + 380 \cdot x_{3,4} + 550 \cdot x_{4,4} + \\ & 145 \cdot x_{1,5} + 235 \cdot x_{2,5} + 320 \cdot x_{3,5} + 460 \cdot x_{4,5} + \end{aligned} \tag{8}$$

Subject to:

functional constraints:

(1) Equality constraint and minimum required capacity

$$\begin{aligned}
P_1 &= 750 + 10\cdot x_{1,1} + 25\cdot x_{2,1} + 50\cdot x_{3,1} + 100\cdot x_{4,1} \le 780\\
P_2 &= \quad P_1 + 10\cdot x_{1,2} + 25\cdot x_{2,2} + 50\cdot x_{3,2} + 100\cdot x_{4,2} \le 860\\
P_3 &= \quad P_2 + 10\cdot x_{1,3} + 25\cdot x_{2,3} + 50\cdot x_{3,3} + 100\cdot x_{4,3} \le 950\\
P_4 &= \quad P_3 + 10\cdot x_{1,4} + 25\cdot x_{2,4} + 50\cdot x_{3,4} + 100\cdot x_{4,4} \le 1'060\\
P_5 &= \quad P_4 + 10\cdot x_{1,5} + 25\cdot x_{2,5} + 50\cdot x_{3,5} + 100\cdot x_{4,5} \le 1'180
\end{aligned} \tag{9}$$

Non-negativity constraints

$$x_{i,t} \ge 0 \tag{10}$$

Integer constraint

$$x_{i,t}, \text{int}$$

3.2.2 Solver

3.2.2.1 Model

```
#Model file: IMM-power-genenator.mod
set GENERATOR;
param T; # No. of years
param P;   # Initial capacity (MW)
param Q {t in 1..T} >= 0;   # minimum yearly required capacity
param c{i in GENERATOR,t in 1..T}; # Yearly generating cost (CHF)
param G{i in GENERATOR}; # Capacity of generator i (MW)
var x {i in GENERATOR}: integer;  # No. of generator to buy
var H {t in 1..T} = (if t == 1 then P+ sum{i in GENERATOR} x[i]*G[i] else H[t-1] + sum{i in GENERATOR}
x[i]*G[i];
minimize purchasingcost:
   sum {i in GENERATOR,t in 1..T} c[i,t] * x[i];
subject to minimumcapacity {t in 1..T}: H[t] >= Q[t];
```

3.2.2.2 Data

```
#Data file: IMM-power-genenator.dat
data;
set GENERATOR:= 10MW 25MW 50MW 100MW;
param T:=5; #planning term (5 years)
#cost of operation and maintenance(CHF/year)
param c :=
          [*,*]:        1      2      3      4      5      :=
          10MW          300    250    200    170    145
          25MW          460    375    350    280    235
          50MW          670    558    465    380    320
          100MW         950    790    670    550    460;
param Q :=  1 780  2 860   3 950    4 1060 5 1180;
param G :=  10MW 10        25MW 25         50MW 50        100MW 100;
param P:= 750;
```

3.2.3 Results

Table 4. Number of generators to be bought in each year and their costs

Generators	Number of generators to be bought in year t				
	t=1	t=2	t=3	t=4	t=5
10 MW	0	1	0	0	0
25 MW	0	0	0	0	1
50 MW	0	0	0	0	0
100 MW	1	0	1	1	1
	Cost of buying generator in year t				
	t=1	t=2	t=3	t=4	t=5
10 MW	0	250	0	0	0
25 MW	0	0	0	0	235
50 MW	0	0	0	0	0
100 MW	950	0	670	550	460
Sub total	**950**	**250**	**670**	**550**	**695**
Total	**3’115**				

3.2.3.1 Results from solver

```
Solved in 2 iterations and 0.00 seconds
Optimal objective  3115
Gurobi 5.5.0: optimal solution; objective 3115
x [*,*] :=      1     2     3     4     5     :=

10MW            0     1     0     0     0
25MW            0     0     0     0     1
50MW            0     0     0     0     0
100MW           1     0     1     1     1;
```

4 ILO - Snow removal problem

Snow removal and disposal are important and expensive activities in your city. Even though snow can be cleared from streets and sidewalks by plowing and shoveling, in prolonged sub-freezing temperatures, the resulting banks of accumulated snow can impede pedestrian and vehicular traffic and must be removed.

To allow timely removal and disposal of snow, you have divided your city into several sectors and snow removal operations are carried out concurrently in each sector.

Accumulated snow is loaded onto trucks and hauled away to disposal sites, e.g., rivers, quarries, sewers, and surface areas. The different types of disposal sites can accommodate different amounts of snow, either because of the physical size of the disposal facility, or because of the environmental restrictions on the amount of snow (often contaminated by salt and de-icing chemicals). The annual capacities for the five different snow disposal sites available to you are given Table 5.

Table 5: Storage capacity of disposal sites

Disposal sites[1]				
1	**2**	**3**	**4**	**5**
350	250	500	400	200

[1]in 1'000 m^3

The cost of removing and disposing of snow depends mainly on the distance it must be trucked. You have decided to use the straight line distance between the centers of each sector to each of the various disposal sites as an approximation of the cost involved in transporting snow between these locations. These distances are given in Table 6.

Table 6. Distance from each sector to each disposal site

sector	Disposal site				
	1	2	3	4	5
1	3.4	1.4	4.9	7.4	9.3
2	2.4	2.1	8.3	9.1	8.8
3	1.4	2.9	3.7	9.4	8.6
4	2.6	3.6	4.5	8.2	8.9
5	1.5	3.1	2.1	7.9	8.8
6	4.2	4.9	6.5	7.7	6.1
7	4.8	6.2	9.9	6.2	5.7
8	5.4	6.0	5.2	7.6	4.9
9	3.1	4.1	6.6	7.5	7.2
10	3.2	6.5	7.1	6.0	8.3

[1]in km

You have estimated, using past snow fall records, that the annual volume of snow requiring removal in each sector as four times the length of the roads in the sectors in meters (i.e. it is assumed that each meter of road generates four cubic meters of snow to remove over the entire year). It costs 0.10 mu to transport 1 $m^3$of snow one kilometer.

Table 7. Quantity of snow to be removed per year

Sector	1	2	3	4	5	6	7	8	9	10
Snow to be removed	153	152	154	138	127	129	111	110	130	135

[1]in m^3

4.1 Question A

What is the most efficient way to remove snow? and how much will it cost?

4.2 Question B

If there were no constraints on the amount of snow that could be dumped at each site what would the most efficient way be to remove snow? and how much would it cost?

4.3 Question C

If you could increase the capacity of a single disposal site by 100 m^3, which would it be? How much should you be willing to pay to do this?

4.4 Answer A

4.4.1 Mathematical formulation

4.4.1.1 Specific

Objective function:

Minimize total cost:

$$Z = 0.1 \cdot 1'000 \cdot \begin{bmatrix} 153(3.4 \cdot x_{1,1} + 1.4 \cdot x_{1,2} + 4.9 \cdot x_{1,3} + 7.4 \cdot x_{1,4} + 9.3 \cdot x_{1,5}) + \\ 152(2.4 \cdot x_{2,1} + 2.1 \cdot x_{2,2} + 8.3 \cdot x_{2,3} + 9.1 \cdot x_{2,4} + 8.8 \cdot x_{2,5}) + \\ 154(1.4 \cdot x_{3,1} + 2.9 \cdot x_{3,2} + 3.7 \cdot x_{3,3} + 9.4 \cdot x_{3,4} + 8.6 \cdot x_{3,5}) + \\ 138(2.6 \cdot x_{4,1} + 3.6 \cdot x_{4,2} + 4.5 \cdot x_{4,3} + 8.2 \cdot x_{4,4} + 8.9 \cdot x_{4,5}) + \\ 127(1.5 \cdot x_{5,1} + 3.1 \cdot x_{5,2} + 2.1 \cdot x_{5,3} + 7.9 \cdot x_{5,4} + 8.8 \cdot x_{5,5}) + \\ 129(4.2 \cdot x_{6,1} + 4.9 \cdot x_{6,2} + 6.5 \cdot x_{6,3} + 7.7 \cdot x_{6,4} + 6.1 \cdot x_{6,5}) + \\ 111(4.8 \cdot x_{7,1} + 6.2 \cdot x_{7,2} + 9.9 \cdot x_{7,3} + 6.2 \cdot x_{7,4} + 5.7 \cdot x_{7,5}) + \\ 110(5.4 \cdot x_{8,1} + 6.0 \cdot x_{8,2} + 5.2 \cdot x_{8,3} + 7.6 \cdot x_{8,4} + 4.9 \cdot x_{8,5}) + \\ 130(3.1 \cdot x_{9,1} + 4.1 \cdot x_{9,2} + 6.6 \cdot x_{9,3} + 7.5 \cdot x_{9,4} + 7.2 \cdot x_{9,5}) + \\ 135(3.2 \cdot x_{10,1} + 6.5 \cdot x_{10,2} + 7.1 \cdot x_{10,3} + 6.0 \cdot x_{10,4} + 8.3 \cdot x_{10,5}) + \end{bmatrix} \tag{11}$$

Subject to:

functional constraints

(1) One sector to one disposal site

$$\begin{bmatrix} x_{1,1}+x_{1,2}+x_{1,3}+x_{1,4}+x_{1,5}=1 \\ x_{2,1}+x_{2,2}+x_{2,3}+x_{2,4}+x_{2,5}=1 \\ x_{3,1}+x_{3,2}+x_{3,3}+x_{3,4}+x_{3,5}=1 \\ x_{4,1}+x_{4,2}+x_{4,3}+x_{4,4}+x_{4,5}=1 \\ x_{5,1}+x_{5,2}+x_{5,3}+x_{5,4}+x_{5,5}=1 \\ x_{6,1}+x_{6,2}+x_{6,3}+x_{6,4}+x_{6,5}=1 \\ x_{7,1}+x_{7,2}+x_{7,3}+x_{7,4}+x_{7,5}=1 \\ x_{8,1}+x_{8,2}+x_{8,3}+x_{8,4}+x_{8,5}=1 \\ x_{9,1}+x_{9,2}+x_{9,3}+x_{9,4}+x_{9,5}=1 \\ x_{10,1}+x_{10,2}+x_{10,3}+x_{10,4}+x_{10,5}=1 \end{bmatrix} \quad (12)$$

(2) Disposal site capacity:

$$\begin{bmatrix} 153\cdot x_{1,1}+152\cdot x_{2,1}+154\cdot x_{3,1}+138\cdot x_{4,1}+127x_{5,1}+129\cdot x_{6,1}+111\cdot x_{7,1}+110\cdot x_{8,1}+130\cdot x_{9,1}+135\cdot x_{10,1}\le 350 \\ 153\cdot x_{1,2}+152\cdot x_{2,2}+154\cdot x_{3,2}+138\cdot x_{4,2}+127x_{5,2}+129\cdot x_{6,2}+111\cdot x_{7,2}+110\cdot x_{8,2}+130\cdot x_{9,2}+135\cdot x_{10,2}\le 250 \\ 153\cdot x_{1,3}+152\cdot x_{2,3}+154\cdot x_{3,3}+138\cdot x_{4,3}+127x_{5,3}+129\cdot x_{6,3}+111\cdot x_{7,3}+110\cdot x_{8,3}+130\cdot x_{9,3}+135\cdot x_{10,3}\le 500 \\ 153\cdot x_{1,4}+152\cdot x_{2,4}+154\cdot x_{3,4}+138\cdot x_{4,4}+127x_{5,4}+129\cdot x_{6,4}+111\cdot x_{7,4}+110\cdot x_{8,4}+130\cdot x_{9,4}+135\cdot x_{10,4}\le 400 \\ 153\cdot x_{1,5}+152\cdot x_{2,5}+154\cdot x_{3,5}+138\cdot x_{4,5}+127x_{5,5}+129\cdot x_{6,5}+111\cdot x_{7,5}+110\cdot x_{8,5}+130\cdot x_{9,5}+135\cdot x_{10,5}\le 200 \end{bmatrix} \quad (13)$$

4.4.2 Solver

4.4.2.1 Model

```
#Model file: IMM-snow-remove-planning-A+B.mod
set SECTOR ordered;
set SITE   ordered;
param s {SECTOR} >= 0; #snow amount m3
param q {SITE} >= 0; #capacity m3
param l {SECTOR, SITE} >= 0; #distance in km
# Decision variables
var x {SECTOR, SITE} binary;
# Objective function
minimize totalCost:
   sum {i in SECTOR, j in SITE}
        1000 * 0.1 * s[i] * l[i,j] * x[i,j];
subject to oneSite {i in SECTOR}:
   sum {j in SITE} x[i,j] = 1;
subject to siteq {j in SITE}:
   sum {i in SECTOR} s[i]*x[i,j] <= q[j];
```

4.4.2.2 Data

```
#Data file: IMM-snow-remove-planning-A+B.dat
data;
param: SECTOR:  s := # defines set "SECTOR" and param "s"
  Sec1 153 Sec2 152 Sec3 154 Sec4 138 Sec5 127
  Sec6 129 Sec7 111 Sec8 110 Sec9 130 Sec10 135 ;
param: SITE:  q := # defines "SITE" and "q"
  Site1 350 Site2 250 Site3 500 Site4 400 Site5 200 ;
param l:
      Site1   Site2   Site3   Site4   Site5 :=
 Sec1  3.4 1.4 4.9 7.4 9.3
 Sec2  2.4 2.1 8.3 9.1 8.8
```

```
Sec3  1.4 2.9 3.7 9.4 8.6
Sec4  2.6 3.6 4.5 8.2 8.9
Sec5  1.5 3.1 2.1 7.9 8.8
Sec6  4.2 4.9 6.5 7.7 6.1
Sec7  4.8 6.2 9.9 6.2 5.7
Sec8  5.4 6.0 5.2 7.6 4.9
Sec9  3.1 4.1 6.6 7.5 7.2
Sec10 3.2 6.5 7.1 6.0 8.3 ;
```

4.4.3 Results

The most efficient way to remove snow is to follow the strategy shown in Table 8. The cost of this strategy is 547’000 *mus*.

Table 8. Optimal snow removal strategy with constraints

Sector	Disposal site				
	1	2	3	4	5
1	0	1	0	0	0
2	1	0	0	0	0
3	0	0	1	0	0
4	0	0	1	0	0
5	0	0	1	0	0
6	0	0	0	1	0
7	0	0	0	1	0
8	0	0	0	0	1
9	1	0	0	0	0
10	0	0	0	1	0

4.4.3.1 Results

```
Optimal solution found (tolerance 1.00e-04)
Best objective 5.470000000000e+05, best bound 5.470000000000e+05, gap 0.0%
Optimize a model with 15 rows, 50 columns and 100 nonzeros
Iteration    Objective       Primal Inf.    Dual Inf.      Time
       0    5.4700000e+05   0.000000e+00   0.000000e+00      0s

Solved in 0 iterations and 0.00 seconds
Optimal objective  5.470000000e+05
Gurobi 5.5.0: optimal solution; objective 547000
32 simplex iterations
x [*,*]
:   Site1 Site2 Site3 Site4 Site5    :=
Sec1    0    1    0    0    0
Sec2    1    0    0    0    0
Sec3    0    0    1    0    0
Sec4    0    0    1    0    0
Sec5    0    0    1    0    0
Sec6    0    0    0    1    0
Sec7    0    0    0    1    0
Sec8    0    0    0    0    1
Sec9    1    0    0    0    0
Sec10   0    0    0    1    0
;
```

4.5 Answer B

If there were no capacity restrictions at the disposal sites, the constraints shown in equation **Error! Reference source not found.**) would be removed from the model’s formulation. The AMPL code is rewritten

4.5.1 Solver

4.5.1.1 Model

```
#Model file: IMM-snow-remove-planning-C.mod
set SECTOR ordered;
set SITE   ordered;
param s {SECTOR} >= 0; #snow amount m3
param q {SITE} >= 0; #capacity m3
param l {SECTOR, SITE} >= 0; #distance in km
# Decision variables
var x {SECTOR, SITE} binary;
# Objective function
minimize totalCost:
    sum {i in SECTOR, j in SITE}
        1000 * 0.1 * s[i] * l[i,j] * x[i,j];
subject to oneSite {i in SECTOR}:
    sum {j in SITE} x[i,j] = 1;
```

4.5.1.2 Data

```
#Data file: IMM-snow-remove-planning-C.dat
data;
param: SECTOR:  s :=  # defines set "SECTOR" and param "s"
  Sec1 153 Sec2 152 Sec3 154 Sec4 138 Sec5 127
  Sec6 129 Sec7 111 Sec8 110 Sec9 130 Sec10 135 ;
param: SITE:  q :=  # defines "SITE" and "q"
  Site1 350 Site2 250 Site3 500 Site4 400 Site5 200 ;
param l:
      Site1   Site2   Site3   Site4   Site5 :=
 Sec1  3.4 1.4 4.9 7.4 9.3
 Sec2  2.4 2.1 8.3 9.1 8.8
 Sec3  1.4 2.9 3.7 9.4 8.6
 Sec4  2.6 3.6 4.5 8.2 8.9
 Sec5  1.5 3.1 2.1 7.9 8.8
 Sec6  4.2 4.9 6.5 7.7 6.1
 Sec7  4.8 6.2 9.9 6.2 5.7
 Sec8  5.4 6.0 5.2 7.6 4.9
 Sec9  3.1 4.1 6.6 7.5 7.2
 Sec10 3.2 6.5 7.1 6.0 8.3 ;
```

4.5.2 Results

The most efficient way to remove snow would then be to follow the strategy shown in Table 9. The cost of this strategy is 374’690 *mus*, a reduction of 172’400 *mus*.

Table 9: Optimal snow removal strategy without constraints

Sector	Disposal site				
	1	2	3	4	5
1	0.0	1.0	0.0	0.0	0.0
2	0.0	1.0	0.0	0.0	0.0
3	1.0	0.0	0.0	0.0	0.0
4	1.0	0.0	0.0	0.0	0.0
5	1.0	0.0	0.0	0.0	0.0
6	1.0	0.0	0.0	0.0	0.0
7	1.0	0.0	0.0	0.0	0.0
8	0.0	0.0	0.0	0.0	1.0
9	1.0	0.0	0.0	0.0	0.0
10	1.0	0.0	0.0	0.0	0.0

4.5.2.1 Results from solver

```
Optimal solution found (tolerance 1.00e-04)
Best objective 3.746900000000e+05, best bound 3.746900000000e+05, gap 0.0%
Optimize a model with 10 rows, 50 columns and 50 nonzeros
Iteration    Objective       Primal Inf.    Dual Inf.      Time
       0    3.7469000e+05   0.000000e+00   0.000000e+00      0s

Solved in 0 iterations and 0.00 seconds
Optimal objective  3.746900000e+05
Gurobi 5.5.0: optimal solution; objective 374690
x [*,*]
:    Site1 Site2 Site3 Site4 Site5    :=
Sec1    0    1    0    0    0
Sec2    0    1    0    0    0
Sec3    1    0    0    0    0
Sec4    1    0    0    0    0
Sec5    1    0    0    0    0
Sec6    1    0    0    0    0
Sec7    1    0    0    0    0
Sec8    0    0    0    0    1
Sec9    1    0    0    0    0
Sec10   1    0    0    0    0
;
```

4.6 Answer C

To solve this question, the model needs to be extended as follows

4.6.1 Mathematical formulation

4.6.1.1 Specific

Objective function:

Minimize total cost:

$$Z = 0.1 \cdot 1'000 \cdot \begin{bmatrix} 153\left(3.4 \cdot x_{1,1} + 1.4 \cdot x_{1,2} + 4.9 \cdot x_{1,3} + 7.4 \cdot x_{1,4} + 9.3 \cdot x_{1,5}\right) + \\ 152\left(2.4 \cdot x_{2,1} + 2.1 \cdot x_{2,2} + 8.3 \cdot x_{2,3} + 9.1 \cdot x_{2,4} + 8.8 \cdot x_{2,5}\right) + \\ 154\left(1.4 \cdot x_{3,1} + 2.9 \cdot x_{3,2} + 3.7 \cdot x_{3,3} + 9.4 \cdot x_{3,4} + 8.6 \cdot x_{3,5}\right) + \\ 138\left(2.6 \cdot x_{4,1} + 3.6 \cdot x_{4,2} + 4.5 \cdot x_{4,3} + 8.2 \cdot x_{4,4} + 8.9 \cdot x_{4,5}\right) + \\ 127\left(1.5 \cdot x_{5,1} + 3.1 \cdot x_{5,2} + 2.1 \cdot x_{5,3} + 7.9 \cdot x_{5,4} + 8.8 \cdot x_{5,5}\right) + \\ 129\left(4.2 \cdot x_{6,1} + 4.9 \cdot x_{6,2} + 6.5 \cdot x_{6,3} + 7.7 \cdot x_{6,4} + 6.1 \cdot x_{6,5}\right) + \\ 111\left(4.8 \cdot x_{7,1} + 6.2 \cdot x_{7,2} + 9.9 \cdot x_{7,3} + 6.2 \cdot x_{7,4} + 5.7 \cdot x_{7,5}\right) + \\ 110\left(5.4 \cdot x_{8,1} + 6.0 \cdot x_{8,2} + 5.2 \cdot x_{8,3} + 7.6 \cdot x_{8,4} + 4.9 \cdot x_{8,5}\right) + \\ 130\left(3.1 \cdot x_{9,1} + 4.1 \cdot x_{9,2} + 6.6 \cdot x_{9,3} + 7.5 \cdot x_{9,4} + 7.2 \cdot x_{9,5}\right) + \\ 135\left(3.2 \cdot x_{10,1} + 6.5 \cdot x_{10,2} + 7.1 \cdot x_{10,3} + 6.0 \cdot x_{10,4} + 8.3 \cdot x_{10,5}\right) + \end{bmatrix} \tag{14}$$

Subject to:

functional constraints

(1) One sector to one disposal site

$$\begin{bmatrix} x_{1,1}+x_{1,2}+x_{1,3}+x_{1,4}+x_{1,5}=1 \\ x_{2,1}+x_{2,2}+x_{2,3}+x_{2,4}+x_{2,5}=1 \\ x_{3,1}+x_{3,2}+x_{3,3}+x_{3,4}+x_{3,5}=1 \\ x_{4,1}+x_{4,2}+x_{4,3}+x_{4,4}+x_{4,5}=1 \\ x_{5,1}+x_{5,2}+x_{5,3}+x_{5,4}+x_{5,5}=1 \\ x_{6,1}+x_{6,2}+x_{6,3}+x_{6,4}+x_{6,5}=1 \\ x_{7,1}+x_{7,2}+x_{7,3}+x_{7,4}+x_{7,5}=1 \\ x_{8,1}+x_{8,2}+x_{8,3}+x_{8,4}+x_{8,5}=1 \\ x_{9,1}+x_{9,2}+x_{9,3}+x_{9,4}+x_{9,5}=1 \\ x_{10,1}+x_{10,2}+x_{10,3}+x_{10,4}+x_{10,5}=1 \end{bmatrix} \tag{15}$$

(2) Disposal site capacity:

$$\begin{bmatrix} 153\cdot x_{1,1}+152\cdot x_{2,1}+154\cdot x_{3,1}+138\cdot x_{4,1}+127x_{5,1}+129\cdot x_{6,1}+111\cdot x_{7,1}+110\cdot x_{8,1}+130\cdot x_{9,1}+135\cdot x_{10,1}\le 350+100\cdot y_1 \\ 153\cdot x_{1,2}+152\cdot x_{2,2}+154\cdot x_{3,2}+138\cdot x_{4,2}+127x_{5,2}+129\cdot x_{6,2}+111\cdot x_{7,2}+110\cdot x_{8,2}+130\cdot x_{9,2}+135\cdot x_{10,2}\le 250+100\cdot y_2 \\ 153\cdot x_{1,3}+152\cdot x_{2,3}+154\cdot x_{3,3}+138\cdot x_{4,3}+127x_{5,3}+129\cdot x_{6,3}+111\cdot x_{7,3}+110\cdot x_{8,3}+130\cdot x_{9,3}+135\cdot x_{10,3}\le 500+100\cdot y_3 \\ 153\cdot x_{1,4}+152\cdot x_{2,4}+154\cdot x_{3,4}+138\cdot x_{4,4}+127x_{5,4}+129\cdot x_{6,4}+111\cdot x_{7,4}+110\cdot x_{8,4}+130\cdot x_{9,4}+135\cdot x_{10,4}\le 400+100\cdot y_4 \\ 153\cdot x_{1,5}+152\cdot x_{2,5}+154\cdot x_{3,5}+138\cdot x_{4,5}+127x_{5,5}+129\cdot x_{6,5}+111\cdot x_{7,5}+110\cdot x_{8,5}+130\cdot x_{9,5}+135\cdot x_{10,5}\le 200+100\cdot y_5 \end{bmatrix} \tag{16}$$

(3) Sites to be expanded $y_1+y_2+y_3+y_4+y_5\le 1$ (17)

4.6.2 Solver

4.6.2.1 Model

```
#Model file: IMM-snow-remove-planning-D.mod
set SECTOR ordered;
set SITE   ordered;
param s {SECTOR} >= 0; #snowAmount
param q {SITE} >= 0; #capacity
param l {SECTOR, SITE} >= 0; #distance
param e >= 0; #expansionSize
param n integer, >= 0;
### --------------------------------------------------
### Decision variables
### --------------------------------------------------
var x {SECTOR, SITE} binary;
var y {SITE} binary;
### --------------------------------------------------
### Objective function
### --------------------------------------------------
minimize totalCost:
   sum {i in SECTOR, j in SITE}       1000 * 0.1 * s[i] * l[i,j] * x[i,j];
### ---------------------------------------------------------------
### Constraints (binary constraints already defined with variables)
### ---------------------------------------------------------------
subject to oneSite {i in SECTOR}:
   sum {j in SITE} x[i,j] = 1;
subject to siteCapacity {j in SITE}:    sum {i in SECTOR} s[i]*x[i,j]     <= q[j] + e*y[j];
subject to numExpansions:    sum {j in SITE} y[j] <= n;
```

4.6.2.2 Data

```
#Data file: IMM-snow-remove-planning-D.dat
```

```
data;
param: SECTOR:  s :=  # defines set "SECTOR" and param "snowAmount"
   Sec1 153 Sec2 152 Sec3 154 Sec4 138 Sec5 127
   Sec6 129 Sec7 111 Sec8 110 Sec9 130 Sec10 135 ;
param: SITE:  q :=  # defines "SITE" and "capacity"
   Site1 350 Site2 250 Site3 500 Site4 400 Site5 200 ;
param l:
      Site1   Site2   Site3   Site4   Site5 :=
 Sec1  3.4 1.4 4.9 7.4 9.3
 Sec2  2.4 2.1 8.3 9.1 8.8
 Sec3  1.4 2.9 3.7 9.4 8.6
 Sec4  2.6 3.6 4.5 8.2 8.9
 Sec5  1.5 3.1 2.1 7.9 8.8
 Sec6  4.2 4.9 6.5 7.7 6.1
 Sec7  4.8 6.2 9.9 6.2 5.7
 Sec8  5.4 6.0 5.2 7.6 4.9
 Sec9  3.1 4.1 6.6 7.5 7.2
 Sec10 3.2 6.5 7.1 6.0 8.3 ;
param e := 100 ;
param n   := 1   ;
```

4.6.3 Results

The most efficient way to remove snow would then be to follow the strategy shown in Table 10. The cost of this strategy is 489’680 *mus*, a reduction of 57’320 *mus*. You should be willing to pay up to this amount / year to expand site 2 by 100 m^3.

Table 10: Optimal snow removal strategy with constraints

sector	Disposal site $x_{i,j}$				
	1	2	3	4	5
1	0	1	0	0	0
2	0	1	0	0	0
3	1	0	0	0	0
4	0	0	1	0	0
5	0	0	1	0	0
6	0	0	0	0	1
7	0	0	0	1	0
8	0	0	1	0	0
9	1	0	0	0	0
10	0	0	0	1	0
Expanded site y_j	0	1	0	0	0

4.6.3.1 Results in solver

```
Optimal solution found (tolerance 1.00e-04)
Best objective 4.896800000000e+05, best bound 4.896800000000e+05, gap 0.0%
Optimize a model with 16 rows, 55 columns and 110 nonzeros
Iteration    Objective       Primal Inf.    Dual Inf.      Time
      0    4.8968000e+05   0.000000e+00   0.000000e+00      0s

Solved in 0 iterations and 0.00 seconds
Optimal objective  4.896800000e+05
Gurobi 5.5.0: optimal solution; objective 489680
26 simplex iterations
x [*,*]
```

```
:    Site1 Site2 Site3 Site4 Site5    :=
Sec1    0    1    0    0    0
Sec2    0    1    0    0    0
Sec3    1    0    0    0    0
Sec4    0    0    1    0    0
Sec5    0    0    1    0    0
Sec6    0    0    0    0    1
Sec7    0    0    0    1    0
Sec8    0    0    1    0    0
Sec9    1    0    0    0    0
Sec10   0    0    0    1    0
;

y [*] :=
Site1  0
Site2  1
Site3  0
Site4  0
Site5  0
;
```

5 References

Larson, R., 2013. *Elementary Linear Algebra, 7th Edition*, Cengage Learning. Available at: http://college.cengage.com/mathematics/larson/elementary_linear/4e/shared/downloads/c09s3.pdf.

NEOS, 2014. NEOS Optimization Guide | NEOS. Available at: http://neos-guide.org/Optimization-Guide [Accessed April 9, 2014].

Ragsdale, C.T., (2008), Spreadsheet Modelling and decision analysis: A practical introduction to management science, 5th edition, 820 pages.